```
(* We find the global extrema of 2 x^2+y^2  on the region x^2+y^2≤4 *)

f = 2 x^2 + y^2
Plot3D[f, {x, -2, 2}, {y, -2, 2}, RegionFunction → Function[{x, y}, x^2 + y^2 ≤ 4]]
ContourPlot[f, {x, -2, 2}, {y, -2, 2}, RegionFunction → Function[{x, y}, x^2 + y^2 ≤ 4]]
```

$2\,x^2 + y^2$





```
Gf = Grad[f, {x, y}]
VectorPlot[Gf, {x, -2, 2}, {y, -2, 2},
 RegionFunction → Function[{x, y}, x^2 + y^2 ≤ 4]]
```

```
{4 x, 2 y}
```




```
(* Critical points *)
crits = Solve[Gf == {0, 0}, {x, y}]
```

```
{{x → 0, y → 0}}
```

```
H = D[Gf, {{x, y}}] /. crits[[1]];
MatrixForm[H]
Det[H]
(* The only critical point inside the region is a local minimum.  *)
f /. {x → 0, y → 0}
```

$$\begin{pmatrix} 4 & 0 \\ 0 & 2 \end{pmatrix}$$

```
8
```

```
0
```

```
(* Boundary *)
g = f /. {x → 2 Cos[t], y → 2 Sin[t]}
Plot[g, {t, 0, 2 Pi}]
gcrits = Solve[{D[g, t] == 0, 0 <= t <= 2 Pi}, t ]
g /. gcrits[[1]]
g /. gcrits[[2]]
g /. gcrits[[3]]
g /. gcrits[[4]]
```

$$8 \text{Cos}[t]^2 + 4 \text{Sin}[t]^2$$



$$\left\{\{t \to 0\}, \left\{t \to \frac{\pi}{2}\right\}, \{t \to \pi\}, \left\{t \to \frac{3\pi}{2}\right\}, \{t \to 2\pi\}\right\}$$

8

4

8

4

```
(* Global ninimum of 0 at the critical point (0,0) *)
(* Global maximum of 8 at two point on the boundary *)
{2 Cos[t], 2 Sin[t]} /. gcrits[[1]]
{2 Cos[t], 2 Sin[t]} /. gcrits[[3]]
```

{2, 0}

{-2, 0}